


Introduzione a *gnuplot*

Il programma può essere trovato a partire dal sito ufficiale <http://www.gnuplot.info>. Nel seguito mostriamo alcuni esempi che possono essere utili per il corso di laboratorio. I commenti sono indicati con un #.


Plot di funzioni semplici

```
set samples 50 # indica che vogliamo 50 campionamenti delle funzioni
plot [-10:10] sin(x),atan(x),cos(atan(x)) # [-10:10] e' il range ascisse
```



Plot della funzione del reticolo di diffrazione: 2 e 10 sorgenti


```
f(x,N)=sin(N*x/2)**2/sin(x/2)**2/N**2
plot f(x,2),f(x,10)
```



aumentiamo il campionamento e cambiamo il range per vedere piu' ordini:

```
set samples 300
set xrange [-3*pi:3*pi]
```

```
replot
```




Plot di un file dati

Nel file "misura.dat" mettiamo i risultati di una misura:


```
# Contenuto del file misura.dat
# Notare che i commenti sono indicati da #
# I commenti servono per ricordare cosa c'e' nel file:
# as es: prima colonna x, seconda colonna y terza colonna Dx,
# e quarta colonna Dy
0.9 1.1 0.2 0.15
1.3 1.5 0.25 0.2
1.7 1.8 0.2 0.25
2.0 1.99 0.3 0.3
```

possiamo fare il plot con una delle seguenti istruzioni:


```
plot 'misura.dat' # sottinteso la x nella prima colonna, la y nella seconda
```



```
plot 'misura.dat' using 1:2:4 with yerr # solo errori y
```




```
plot 'misura.dat' with xerrorbars # errori x e errori y
```



Alcuni miglioramenti

```
set title "caratteristica I-V" # titolo
set xlabel "corrente I (mA)" # Label asse x
set ylabel "tensione V (Volt)" # label asse y
set xrange [0.5:2.25] # cambia range x
set yrange [0.5:2.5] # cambia range y
set xtics (0.8,1.2,1.6,2) # tic-marks
set label "punto interessante" at 1.0, 2.2 # label
replot # riaggiorna il grafico
```



Altri comandi utili:


```
set autoscale # scala automatica
set logscale # scala log
unset logscale; set logscale y # scala log solo su y
unset xtics; set xtics auto # tics automatici
```

In generale per lavorare con i parametri si usano le istruzioni:

```
show parameter # mostra l'impostazione di parameter
show all # mostra tutte le impostazioni
set parameter value # imposta il parametro
unset parameter # disabilita il parametro
```

Barre di errori e curva teorica

```
set title "misura con curva teorica"
set xlabel "corrente [mA]"
set ylabel "tensione [V]"
V(x,R)=R*x
plot 'misura.dat' title "Misure" with xerrorbars, V(x,1) title "legge di Ohm"
w lines
```



Esportazione dei grafici

Con il seguente comando si vedono i possibili tipi di output:

```
set terminal
```

ad es. si può ottenere:

Available terminal types:

```
corel EPS format for CorelDRAW
dumb ascii art for anything that prints text
dxf dxf-file for AutoCad (default size 120x80)
gif GIF images using libgd and TrueType fonts
jpeg JPEG images using libgd and TrueType fonts
latex LaTeX picture environment
png PNG images using libgd and TrueType fonts
...
```

ad es. si sceglie il formato png, e come file di uscita 'fig1.png':

```
set terminal png size 1000,800
set output 'fig1.png'
```

a questo punto un comando grafico (ad es. replot) va a generare il grafico nel file scelto.

Salvataggio della sessione grafica

Dopo aver fatto un grafico si possono salvare i comandi che si sono impiegati con l'istruzione

```
save "registrazione.plt"
```

Successivamente si possono richiamare tali comandi con

```
load "registrazione.plt"
```